Необходимо установить отбор по периоду для динамического списка формы журнала документов при открытии формы. Надо, чтоб отбор отрабатывал и отображался так как при ручном выборе периода. Пробовал тремя способами задать - ни один нормально не отрабатывает:

```1c
    ДатаНачала = ТекущаяДатаСеанса();//НачалоМесяца(НачалоМесяца(ТекущаяДатаСеанса())-1);
    ДатаОкончания = ТекущаяДатаСеанса();//КонецМесяца(ТекущаяДатаСеанса());
    //1.
    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбораДинамическогоСписка(Список,"Дата",ДатаНачала, ВидСравненияКомпоновкиДанных.БольшеИлиРавно,,Истина);
    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбораДинамическогоСписка(Список,"Дата",ДатаОкончания, ВидСравненияКомпоновкиДанных.МеньшеИлиРавно,,Истина);
    //2.
    ПользовательскийОтбор = Список.КомпоновщикНастроек.ПользовательскиеНастройки.Элементы.Найти(
    Список.КомпоновщикНастроек.Настройки.Отбор.ИдентификаторПользовательскойНастройки);
    УсловиеОтбора = ПользовательскийОтбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    УсловиеОтбора.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("Дата");
    УсловиеОтбора.ВидСравнения = ВидСравненияКомпоновкиДанных.Больше;
    УсловиеОтбора.ПравоеЗначение = ТекущаяДата();
    УсловиеОтбора.Использование = Истина;
    //3.
    УсловиеОтбора = Список.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    УсловиеОтбора.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("Дата");
    УсловиеОтбора.ВидСравнения = ВидСравненияКомпоновкиДанных.Больше;
    УсловиеОтбора.ПравоеЗначение = ТекущаяДата();
    УсловиеОтбора.Использование = Истина;
```

Как его правильно задать?
---
```1c
    &НаКлиенте
    Процедура ПриОткрытии(Отказ)

    Период = новый СтандартныйПериод;
    Период.Вариант = ВариантСтандартногоПериода.Сегодня;

    Элементы.Список.Период = Период;
    Элементы.Список.Обновить();

    КонецПроцедуры
```
---
Вот так сработало.

```1c
    Процедура УстановитьПериодПоУмолчанию(Форма) Экспорт
    Список = Форма.Список;
    ДатаНачала = НачалоМесяца(ТекущаяДатаСеанса());
    ДатаНачала = ДобавитьМесяц(ДатаНачала, -1);
    ДатаОкончания = КонецМесяца(ТекущаяДатаСеанса());
    лНовыйПериод = Новый СтандартныйПериод(ДатаНачала, ДатаОкончания);
    ЗаполнитьЗначенияСвойств(Форма.Элементы.Список.Период, лНовыйПериод);
    Форма.Элементы.Список.Обновить();
    КонецПроцедуры
```

Вызвал из "ПриСозданииНаСервере"
---
Если "при открытии" то после установки отборов нужно добавить:
```1c
    Элементы.Список.Обновить();
```

Но получится что "ПриСозданииНаСервере" список сформировался, потом "ПриОткрытии" вы устанавливаете отборы и список повторно формируется но уже с отборами.
Логичнее формировать список "ПриСозданииНаСервере"

```1c
    Список.Параметры.УстановитьЗначениеПараметра("ДатаНачала", НачалоДня(ТекущаяДатаСеанса()));
    Список.Параметры.УстановитьЗначениеПараметра("ДатаОкончания", ТекущаяДатаСеанса());
```
---
С параметрами не работает:
https://disk.yandex.ru/i/NyB6PZa0sAG_sg
Сделал следующим образом:

```1c
    ОбщегоНазначенияКлиентСервер.ДобавитьЭлементКомпоновки(Список.Отбор,"Дата",ВидСравненияКомпоновкиДанных.БольшеИлиРавно, ДатаНачала,,Истина,РежимОтображенияЭлементаНастройкиКомпоновкиДанных.Обычный);
    ОбщегоНазначенияКлиентСервер.ДобавитьЭлементКомпоновки(Список.Отбор,"Дата",ВидСравненияКомпоновкиДанных.МеньшеИлиРавно, ДатаОкончания,,Истина,РежимОтображенияЭлементаНастройкиКомпоновкиДанных.Обычный);
```

 Отбор срабатывает, но в форме не отображается, что установлен отбор по периоду.
---
Сделайте произвольный запрос в свойствах динамического списка с выбором нужных полей и укажите условия отбора по параметрам которые устанавливаете (&ДатаНачала и &ДатаОкончания) тогда ошибки не будет
---
Естественно, вы же не сказали какие параметры есть. Как должен отображаться отбор? Если это кнопка на форме то установить пометку например: Элементы.Период.Пометка = Истина;
---
А если установить РежимОтображенияЭлементаНастройкиКомпоновкиДанных.БыстрыйДоступ?
---
```1c
    &НаКлиенте
    Процедура ПриОткрытии(Отказ)

    Период = новый СтандартныйПериод;
    Период.Вариант = ВариантСтандартногоПериода.Сегодня;

    Элементы.Список.Период = Период;
    Элементы.Список.Обновить();

    КонецПроцедуры
```
---
Недавно сам искал решение подобной задачи. Остановился вот на таком коде:

```1c
    НастройкиКомпоновки=Новый НастройкиКомпоновкиДанных;
    ГруппаОтбора=НастройкиКомпоновки.Отбор.Элементы.Добавить(Тип("ГруппаЭлементовОтбораКомпоновкиДанных"));
    ГруппаОтбора.ТипГруппы=ТипГруппыЭлементовОтбораКомпоновкиДанных.ГруппаИ;
    ЭлементОтбора=НастройкиКомпоновки.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    ЭлементОтбора.Использование=Истина;
    ЭлементОтбора.ЛевоеЗначение=Новый ПолеКомпоновкиДанных("Номенклатура");
    ЭлементОтбора.ВидСравнения=ВидСравненияКомпоновкиДанных.ВСписке;
    ЭлементОтбора.ПравоеЗначение=МассивНоменклатуры;
    Если НЕ Элементы.Таб.ОтборСтрок=Неопределено Тогда
    ЭлементОтбора=ГруппаОтбора.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    ЭлементОтбора.ЛевоеЗначение=Новый ПолеКомпоновкиДанных("Период");
    ЭлементОтбора.ВидСравнения=ВидСравненияКомпоновкиДанных.БольшеИлиРавно;
    ЭлементОтбора.Использование=Истина;
    ЭлементОтбора.ПравоеЗначение=Элементы.Таб.ОтборСтрок.Период;
    ЭлементОтбора=ГруппаОтбора.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    ЭлементОтбора.ЛевоеЗначение=Новый ПолеКомпоновкиДанных("Период");
    ЭлементОтбора.ВидСравнения=ВидСравненияКомпоновкиДанных.МеньшеИлиРавно;
    ЭлементОтбора.Использование=Истина;
    ЭлементОтбора.ПравоеЗначение=КонецДня(Элементы.Таб.ОтборСтрок.Период);
    КонецЕсли;
    ПараметрыВыбора=Новый Структура;
    ПараметрыВыбора.Вставить("ФиксированныеНастройки",НастройкиКомпоновки);
    ПараметрыВыбора.Вставить("РежимВыбора",Истина);
    ПараметрыВыбора.Вставить("МножественныйВыбор",Истина);
    ОписаниеОповещения = Новый ОписаниеОповещения("ПослеВыводаРегистра",ЭтотОбъект,"Подбор");
    ОткрытьФорму("РегистрНакопления.ТоварыНаСкладах.Форма.ФормаСписка",ПараметрыВыбора,ЭтаФорма,УникальныйИдентификатор,,,ОписаниеОповещения);
```

Работает, как мне надо.
---
Вот так сработало.

```1c
    Процедура УстановитьПериодПоУмолчанию(Форма) Экспорт
    Список = Форма.Список;
    ДатаНачала = НачалоМесяца(ТекущаяДатаСеанса());
    ДатаНачала = ДобавитьМесяц(ДатаНачала, -1);
    ДатаОкончания = КонецМесяца(ТекущаяДатаСеанса());
    лНовыйПериод = Новый СтандартныйПериод(ДатаНачала, ДатаОкончания);
    ЗаполнитьЗначенияСвойств(Форма.Элементы.Список.Период, лНовыйПериод);
    Форма.Элементы.Список.Обновить();
    КонецПроцедуры
```

Вызвал из "ПриСозданииНаСервере"